5 次方程式の短い解の公式 A　with Mathematica14 .0

In[1118]:=

```
ClearAll["`*"];
```

このファイル（formulaA.nb）では「Solving Quintics and Septics by Radicals
(Monammed A.Faggal, Daniel Lazard共著)」（以下【原論文】と略記)
で示された「5 次方程式の解の公式 （formulaA)」の説明をします.
これは共著者の一人である Lazard 氏が,
この論文の10年ほど前に発表された「ガロア分解式を使った解の公式」(3 ページ) より,
はるかに短く簡潔です.
（私も自分で「ガロア分解式を使った解の公式 （の様な物）」を作りましたが,
それは5千行ぐらいになりました. 「$a_1 = 0$」を代入するとこちらもずーと短くなると思いますが,
それでも3ページよりは長いと思います.)

なお【原論文】では「7 次方程式の解の公式」も載っていますが,
それは「5 次方程式の解の公式（formulaA)」とは少し異なる方法で導かれています.
その方法も非常に鮮やかなので, それと同様にして,
もう一つの短い「5 次方程式の解の公式（formulaB)」も自作しました.
そちらは「formulaB.nb」をご覧ください. なお,
私のサイトには上記の「7 次方程式の解の公式」の説明と,
さらに「解を求めるプログラム」も置いてあります.

a_1 と a1 など添え字の大きさのみ違う変数は同じ変数を表すものとします.
原則として text の中では読みやすいように a_1 を使い, input 文の中では a1 を使います .)
また, **原則として,** 抽象的な関数は 大文字で, それを具体化した関数は小文字で表すとします.
(これは【原論文】でも同じやり方です.) また変数に関しては具体的な数値には
$a の様に「$」を付けて区別する事もあります.
そして Global変数は「$」を付けることが多いです.

このファイルは Mathematica 14.0 で作成されています.
コマンドの簡単な説明は入れていますが, 詳しい説明は Google してください.
Wolfram のサイトで詳しい説明を例題付きで読むことができます.
また原論文は Maple で書かれています. したがって, 残念ながら,
グレブナー基底に関しては再現できないこともありました.

Mathematica をお持ちの方は「ノートブックを評価」(15秒程度) して頂ければ　対話的 （interactive）に読むことができて, 分かりやすいと思います. お持ちでない方は PDF をご覧になってくれても良いのですが, PDFは文の体裁がかなり崩れ,プログラムは右端が切れてしまいます （ここは是非WolframとAdobeに頑張ってほしい点です). しかし, 無料の Wolfram Player を使えば , 元の体裁のままノートブックを読むことができるのでお勧めです.

§1. F_{10}について

既約でmonicな有理数係数の5次の整式を $f = x^5 + a_1 x^4 + a_2 x^3 + a_3 x^2 + a_4 x + a_5$,
$f = 0$ の解を x_0, x_1, x_2 , x_3, x_4 と置きます.
$k = 0, 1, 2, 3, 4$ に対し「$\sigma : x_k \to x_{k+1}$, $\tau : x_k \to x_{3k}$」(添え数字はMod5) とおくと,
巡回群$C_5 = <\sigma>$, 2 面体群$D_5 = <\sigma, \tau^2>$,
Frobenius群$F_{20} = <\sigma, \tau>$で, 位数はそれぞれ 5, 10, 20 です.
そして fのガロア群をGalとすると, fが可解の時, Galは上の3つのいずれかです.
(σ, τ の定義は, 私の以前のファイルとは異なりますが, 以前の定義と共役です.)

$\{x_0, x_1, x_2, x_3, x_4\}$ のうち異なる2個をとりその和を作ります.
その集合を $Y = \{y_1, y_2 ..., y_{10}\}$ と置きます.

In[1119]:=

```
Y = Plus @@@ Subsets[{x0, x1, x2, x3, x4}, {2}]
```

Out[1119]=

```
{x0 + x1, x0 + x2, x0 + x3, x0 + x4, x1 + x2, x1 + x3, x1 + x4, x2 + x3, x2 + x4, x3 + x4}
```

「$F10 = (x - y_1)(x - y_2) \cdots (x - y_{10})$」と定義します.

In[1120]:=

```
F10 = Product[x - Y[[k]], {k, 1, 10}]
```

Out[1120]=

```
(x - x0 - x1) (x - x0 - x2) (x - x1 - x2) (x - x0 - x3) (x - x1 - x3)
 (x - x2 - x3) (x - x0 - x4) (x - x1 - x4) (x - x2 - x4) (x - x3 - x4)
```

これは vars = $\{x_0, x_1, \cdots, x_4\}$ の対称式なので , vars の基本対称式 s_k で表されます.
更に「解と係数の関係」より $a_1, a_2, \cdots, a_5$ で表せます.

SymmetricReduction[f, $\{x_1, \ldots, x_n\}$]
$x_1, \ldots, x_n$において $f == p + q$ であるような多項式のペア$\{p, q\}$を与える. ただし, p は対称な部分であり q は剰余である.

SymmetricReduction[f, $\{x_1, \ldots, x_n\}$, $\{s_1, \ldots, s_n\}$]
p における初等対称式を $s_1, \ldots, s_n$ で置き換えたペア $\{p, q\}$ を与える.

In[1121]:=

```
vars = {x4, x3, x2, x1, x0};
F10 = SymmetricReduction[F10, vars, {-a1, a2, -a3, a4, -a5}][[1]] // Collect[#, x] &
```

Out[1122]=

$$a1\,a2\,a3\,a4 - a3^2\,a4 - a1^2\,a4^2 - a1\,a2^2\,a5 + a2\,a3\,a5 + 2\,a1\,a4\,a5 - a5^2 + \left(a1\,a2\,a3^2 - a3^3 + a1\,a2^2\,a4 - 4\,a1\,a4^2 - 4\,a1^2\,a2\,a5 - a2^2\,a5 + 4\,a1\,a3\,a5 + 4\,a4\,a5\right) x + \left(2\,a1\,a2^2\,a3 + a1^2\,a3^2 - a2\,a3^2 + a1^2\,a2\,a4 + a2^2\,a4 - 3\,a1\,a3\,a4 - 4\,a4^2 - 4\,a1^3\,a5 - 9\,a1\,a2\,a5 + 7\,a3\,a5\right) x^2 + \left(a1\,a2^3 + 4\,a1^2\,a2\,a3 + a2^2\,a3 - 4\,a3\,a4 - 16\,a1^2\,a5 - 4\,a2\,a5\right) x^3 + \left(3\,a1^2\,a2^2 + a2^3 + 2\,a1^3\,a3 + 6\,a1\,a2\,a3 - a3^2 - 2\,a1^2\,a4 - 2\,a2\,a4 - 22\,a1\,a5\right) x^4 + \left(3\,a1^3\,a2 + 6\,a1\,a2^2 + 5\,a1^2\,a3 + 2\,a2\,a3 - 5\,a1\,a4 - 11\,a5\right) x^5 + \left(a1^4 + 9\,a1^2\,a2 + 3\,a2^2 + 4\,a1\,a3 - 3\,a4\right) x^6 + \left(4\,a1^3 + 9\,a1\,a2 + a3\right) x^7 + \left(6\,a1^2 + 3\,a2\right) x^8 + 4\,a1\,x^9 + x^{10}$$

すなわち

$$F_{10} = \prod_{0\le i<j\le 4} \{[x - (x_i + x_j)\} = x^{10} + 4\,a_1\,x^9 + \left(6\,a_1^2 + 3\,a_2\right) x^8 + \left(4\,a_1^3 + 9\,a_1\,a_2 + a_3\right) x^7 + \left(a_1^4 + 9\,a_1^2\,a_2 + 3\,a_2^2 + 4\,a_1\,a_3 - 3\,a_4\right) x^6 + \left(3\,a_1^3\,a_2 + 6\,a_1\,a_2^2 + 5\,a_1^2\,a_3 + 2\,a_2\,a_3 - 5\,a_1\,a_4 - 11\,a_5\right) x^5 + \left(3\,a_1^2\,a_2^2 + a_2^3 + 2\,a_1^3\,a_3 + 6\,a_1\,a_2\,a_3 - a_3^2 - 2\,a_1^2\,a_4 - 2\,a_2\,a_4 - 22\,a_1\,a_5\right) x^4 + \left(a_1\,a_2^3 + 4\,a_1^2\,a_2\,a_3 + a_2^2\,a_3 - 4\,a_3\,a_4 - 16\,a_1^2\,a_5 - 4\,a_2\,a_5\right) x^3 + \left(2\,a_1\,a_2^2\,a_3 + a_1^2\,a_3^2 - a_2\,a_3^2 + a_1^2\,a_2\,a_4 + a_2^2\,a_4 - 3\,a_1\,a_3\,a_4 - 4\,a_4^2 - 4\,a_1^3\,a_5 - 9\,a_1\,a_2\,a_5 + 7\,a_3\,a_5\right) x^2 + \left(a_1\,a_2\,a_3^2 - a_3^3 + a_1\,a_2^2\,a_4 - 4\,a_1\,a_4^2 - 4\,a_1^2\,a_2\,a_5 - a_2^2\,a_5 + 4\,a_1\,a_3\,a_5 + 4\,a_4\,a_5\right) x + a_1\,a_2\,a_3\,a_4 - a_3^2\,a_4 - a_1^2\,a_4^2 - a_1\,a_2^2\,a_5 + a_2\,a_3\,a_5 + 2\,a_1\,a_4\,a_5 - a_5^2$$

【コメント】原論文では， 終結式 (resultant) を使って導いています .

§2. F_1, F_2について

σ によるYの軌道分解は 次の O_1, O_2 の二つしかありません.
(Mod5 で考えると, O_1 は添え数字の差が1, O_2 は差が2)

In[1124]:=

In[1125]:=

```
O1 = Table[Total[vars[[Sort[Mod[{0, 1} + k, 5]] + 1]]], {k, 0, 4}];
O2 = Table[Total[vars[[Sort[Mod[{0, 2} + k, 5]] + 1]]], {k, 0, 4}];
Column[{"O1:" <> ToString[O1], "O2:" <> ToString[O2]}]
```

Out[1127]=

O_1:{x3 + x4, x2 + x3, x1 + x2, x0 + x1, x0 + x4}
O_2:{x2 + x4, x1 + x3, x0 + x2, x1 + x4, x0 + x3}

O_1, O_2 の要素を解に持つ5次の整式をそれぞれ F_1, F_2 と置きます.

In[1128]:=

```
F1 = Product[(x - O1[[k]]), {k, 1, 5}]
F2 = Product[(x - O2[[k]]), {k, 1, 5}]
```

Out[1128]=

(x - x0 - x1) (x - x1 - x2) (x - x2 - x3) (x - x0 - x4) (x - x3 - x4)

Out[1129]=

(x - x0 - x2) (x - x0 - x3) (x - x1 - x3) (x - x1 - x4) (x - x2 - x4)

F_1, F_2 はσ, τ^2 で変わらず, τによって $F_1 \longleftrightarrow F_2$ と移ります.
また「f = 0」の判別式を D とおくと, $\sqrt{D}$ はσ, τ^2 で変わらず,
τによって $\sqrt{D} \longleftrightarrow -\sqrt{D}$ と移ります. 故にF_1, F_2 の係数は $Q(\sqrt{D})$ に入ります.
また「5 次の整式の根が1つの軌道を形成する場合、その整式は既約多項式または $(x-\alpha)^5$ となる」
ので, $Gal = F_{10}$ またはC$_5$ のとき, F_1, F_2 は共に「Q上既約な5次式」となります.
$Gal = F_{20}$ のとき, F_{10} はQ上既約ですが, $Q(\sqrt{D})$ 上では $F_{10} = F_1 F_2$ と 因数分解されます.

■ 【例1】$f = 7 + 10x - 10x^2 - 15x^3 + x^5$ ($Gal = C_5$)

In[1130]:=

```
f = 7 + 10 x - 10 x^2 - 15 x^3 + x^5;
f10 =
 F10 /. AssociationThread[{a1, a2, a3, a4, a5} → Reverse@CoefficientList[f, x, 5]]
d = Sqrt[Discriminant[f, x]] (*d= √D *)
Factor[f10]
```

Out[1131]=

$1 - 295x + 2860x^2 - 1430x^3 - 3175x^4 + 223x^5 + 645x^6 - 10x^7 - 45x^8 + x^{10}$

Out[1132]=

4375

Out[1133]=

$(-1 + 285x - 5x^2 - 35x^3 + x^5)(-1 + 10x - 5x^2 - 10x^3 + x^5)$

f_{10} はQ上でf_1, f_2 に因数分解できました.

■ 【例2】$f = 4 + 10x - x^2 - 6x^3 + x^5$ $(\mathrm{Gal} = F_{10})$

In[1134]:=

```
f = 4 + 10 x - x^2 - 6 x^3 + x^5;
f10 =
 F10 /. AssociationThread[{a1, a2, a3, a4, a5} → Reverse@CoefficientList[f, x, 5]]
d = Sqrt[Discriminant[f, x]] (*d= √D *)
Factor[f10]
```

Out[1135]=

$$-2 + 17x - 62x^2 + 100x^3 - 97x^4 - 32x^5 + 78x^6 - x^7 - 18x^8 + x^{10}$$

Out[1136]=

262

Out[1137]=

$$\left(-1 + 6x - 11x^2 - 11x^3 + x^5\right)\left(2 - 5x + 10x^2 - 7x^3 + x^5\right)$$

f_{10} はQ上でf_1, f_2 に因数分解できました.

■ 【例3】$f = 12 - 6x + 12x^2 - 9x^3 + x^5$ $(\mathrm{Gal} = F_{20})$

In[1138]:=

```
f = 5 + 5 x - 5 x^3 + x^5
f10 =
 F10 /. AssociationThread[{a1, a2, a3, a4, a5} → Reverse@CoefficientList[f, x, 5]]
Factor[f10];
d = Sqrt[Discriminant[f, x]] (*d= √D *)
$f10 = Factor[f10, Extension → d]
f1 = FactorList[$f10, Extension → d][[3, 1]] * (-1 / 2) // Expand
f2 = FactorList[$f10, Extension → d][[4, 1]] * (1 / 2) // Expand
```

Out[1138]=

$$5 + 5x - 5x^3 + x^5$$

Out[1139]=

$$-25 - 25x + 25x^2 + 100x^3 - 75x^4 - 55x^5 + 60x^6 - 15x^8 + x^{10}$$

Out[1141]=

$$525\sqrt{5}$$

Out[1142]=

$$-\frac{1}{4}\left(55 + 25\sqrt{5} + \left(-35 - 15\sqrt{5}\right)x + \left(15 + 5\sqrt{5}\right)x^3 - 2x^5\right)$$
$$\left(-55 + 25\sqrt{5} + \left(35 - 15\sqrt{5}\right)x + \left(-15 + 5\sqrt{5}\right)x^3 + 2x^5\right)$$

Out[1143]=

$$-\frac{55}{2} - \frac{25\sqrt{5}}{2} + \frac{35x}{2} + \frac{15\sqrt{5}\,x}{2} - \frac{15x^3}{2} - \frac{5\sqrt{5}\,x^3}{2} + x^5$$

Out[1144]=

$$-\frac{55}{2} + \frac{25\sqrt{5}}{2} + \frac{35x}{2} - \frac{15\sqrt{5}\,x}{2} - \frac{15x^3}{2} + \frac{5\sqrt{5}\,x^3}{2} + x^5$$

f_{10} はQ上既約ですが, $Q\left(\sqrt{D}\right)$ 上ではf_1, f_2 に因数分解できました.

f1, f2を見つけるプログラム *findF12* は次の様に作れます. 出力は{f1,f2}で, $f1,$f2 は f1,f2のGlobal変数

です.

In[1145]:=

```
findF12[f_] := Module[{factors, pos},
  Clear[a1, a2, a3, a4, a5];
  $f10 = F10 /.
    AssociationThread[{a1, a2, a3, a4, a5} → Reverse@CoefficientList[f, x, 5]];
  factors = FactorList[$f10, Extension → Sqrt[Discriminant[f, x]]];
   pos = Position[Exponent[factors, x], 5][[All, 1]];
  {$f1, $f2} = Expand[# / Coefficient[#, x, 5]] & /@ factors[[All, 1]][[pos]];
  If[$f1 == x^5, {$f1, $f2} = {$f2, $f1}];
  Return[{$f1, $f2}]
 ]
```

In[1146]:=

(*例*)findF12$\left[5+5x-5x^3+x^5\right]$

Out[1146]=

$$\left\{-\frac{55}{2}-\frac{25\sqrt{5}}{2}+\frac{35x}{2}+\frac{15\sqrt{5}\,x}{2}-\frac{15x^3}{2}-\frac{5\sqrt{5}\,x^3}{2}+x^5,\right.$$
$$\left.-\frac{55}{2}+\frac{25\sqrt{5}}{2}+\frac{35x}{2}-\frac{15\sqrt{5}\,x}{2}-\frac{15x^3}{2}+\frac{5\sqrt{5}\,x^3}{2}+x^5\right\}$$

In[1147]:=

$f1

Out[1147]=

$$-\frac{55}{2}-\frac{25\sqrt{5}}{2}+\frac{35x}{2}+\frac{15\sqrt{5}\,x}{2}-\frac{15x^3}{2}-\frac{5\sqrt{5}\,x^3}{2}+x^5$$

Gal $=F_{20}$ のとき, f1とf2は $\sqrt{D}$ の係数の符号のみ異なります.
 したがって f_1+f_2 , $\sqrt{D}$ (f_1-f_2) の係数は Qに含まれます.

In[1148]:=

```
$f1 + $f2
Sqrt[5] ($f1 - $f2) // Expand
```

Out[1148]=

$-55+35x-15x^3+2x^5$

Out[1149]=

$-125+75x-25x^3$

§3. F^+, F^-について

$F^+ = f_1 + f_2$, $F^- = f_1 - f_2$ と定義します．F^+ と F^- の係数は，それぞれ Q, $Q\left(\sqrt{D}\right)$ に入ります．
さらに F^+, $\sqrt{5}\ F^-$ の（$5-i$）次の係数をそれぞれ e_i, d_i と置きます．（$1 \leq i \leq 5$）
なお，Mathematicaの仕様で $F^+ = Fp$, $F^- = Fm$ と書きます．
　一般の整式fの場合は，d_i, e_i は，$x_0 \sim x_4$ の式になります．

In[1150]:=

```
Fp = F1 + F2
Fm = F1 - F2
```

Out[1150]=

$$(x - x0 - x2)(x - x0 - x3)(x - x1 - x3)(x - x1 - x4)(x - x2 - x4) + (x - x0 - x1)(x - x1 - x2)(x - x2 - x3)(x - x0 - x4)(x - x3 - x4)$$

Out[1151]=

$$-((x - x0 - x2)(x - x0 - x3)(x - x1 - x3)(x - x1 - x4)(x - x2 - x4)) + (x - x0 - x1)(x - x1 - x2)(x - x2 - x3)(x - x0 - x4)(x - x3 - x4)$$

In[1152]:=

```
{e1, e2, e3, e4, e5} = CoefficientList[Fp, x, 5] // Reverse
{d1, d2, d3, d4, d5} = Sqrt[5] CoefficientList[Fm, x, 5] // Reverse
```

Out[1152]=

$$\{-4\,x0-4\,x1-4\,x2-4\,x3-4\,x4,\ 2\,x0^2+7\,x0\,x1+2\,x1^2+7\,x0\,x2+7\,x1\,x2+$$
$$2\,x2^2+7\,x0\,x3+7\,x1\,x3+7\,x2\,x3+2\,x3^2+7\,x0\,x4+7\,x1\,x4+7\,x2\,x4+7\,x3\,x4+2\,x4^2,$$
$$-3\,x0^2\,x1-3\,x0\,x1^2-3\,x0^2\,x2-10\,x0\,x1\,x2-3\,x1^2\,x2-3\,x0\,x2^2-3\,x1\,x2^2-3\,x0^2\,x3-$$
$$10\,x0\,x1\,x3-3\,x1^2\,x3-10\,x0\,x2\,x3-10\,x1\,x2\,x3-3\,x2^2\,x3-3\,x0\,x3^2-3\,x1\,x3^2-$$
$$3\,x2\,x3^2-3\,x0^2\,x4-10\,x0\,x1\,x4-3\,x1^2\,x4-10\,x0\,x2\,x4-10\,x1\,x2\,x4-3\,x2^2\,x4-$$
$$10\,x0\,x3\,x4-10\,x1\,x3\,x4-10\,x2\,x3\,x4-3\,x3^2\,x4-3\,x0\,x4^2-3\,x1\,x4^2-3\,x2\,x4^2-3\,x3\,x4^2,$$
$$x0^2\,x1^2+3\,x0^2\,x1\,x2+4\,x0\,x1^2\,x2+x0^2\,x2^2+3\,x0\,x1\,x2^2+x1^2\,x2^2+3\,x0^2\,x1\,x3+3\,x0\,x1^2\,x3+$$
$$4\,x0^2\,x2\,x3+10\,x0\,x1\,x2\,x3+3\,x1^2\,x2\,x3+3\,x0\,x2^2\,x3+4\,x1\,x2^2\,x3+x0^2\,x3^2+4\,x0\,x1\,x3^2+x1^2\,x3^2+$$
$$3\,x0\,x2\,x3^2+3\,x1\,x2\,x3^2+x2^2\,x3^2+4\,x0^2\,x1\,x4+3\,x0\,x1^2\,x4+3\,x0^2\,x2\,x4+10\,x0\,x1\,x2\,x4+$$
$$3\,x1^2\,x2\,x4+4\,x0\,x2^2\,x4+3\,x1\,x2^2\,x4+3\,x0^2\,x3\,x4+10\,x0\,x1\,x3\,x4+4\,x1^2\,x3\,x4+10\,x0\,x2\,x3\,x4+$$
$$10\,x1\,x2\,x3\,x4+3\,x2^2\,x3\,x4+3\,x0\,x3^2\,x4+3\,x1\,x3^2\,x4+4\,x2\,x3^2\,x4+x0^2\,x4^2+3\,x0\,x1\,x4^2+$$
$$x1^2\,x4^2+3\,x0\,x2\,x4^2+4\,x1\,x2\,x4^2+x2^2\,x4^2+4\,x0\,x3\,x4^2+3\,x1\,x3\,x4^2+3\,x2\,x3\,x4^2+x3^2\,x4^2,$$
$$-x0^2\,x1^2\,x2-x0\,x1^2\,x2^2-2\,x0^2\,x1\,x2\,x3-2\,x0\,x1^2\,x2\,x3-x0^2\,x2^2\,x3-2\,x0\,x1\,x2^2\,x3-$$
$$x1^2\,x2^2\,x3-x0^2\,x1\,x3^2-x0\,x1^2\,x3^2-x0^2\,x2\,x3^2-2\,x0\,x1\,x2\,x3^2-x1\,x2^2\,x3^2-$$
$$x0^2\,x1^2\,x4-2\,x0^2\,x1\,x2\,x4-2\,x0\,x1^2\,x2\,x4-x0^2\,x2^2\,x4-2\,x0\,x1\,x2^2\,x4-2\,x0^2\,x1\,x3\,x4-$$
$$2\,x0\,x1^2\,x3\,x4-2\,x0^2\,x2\,x3\,x4-4\,x0\,x1\,x2\,x3\,x4-2\,x1^2\,x2\,x3\,x4-2\,x0\,x2^2\,x3\,x4-$$
$$2\,x1\,x2^2\,x3\,x4-2\,x0\,x1\,x3^2\,x4-x1^2\,x3^2\,x4-2\,x0\,x2\,x3^2\,x4-2\,x1\,x2\,x3^2\,x4-x2^2\,x3^2\,x4-$$
$$x0^2\,x1\,x4^2-2\,x0\,x1\,x2\,x4^2-x1^2\,x2\,x4^2-x0\,x2^2\,x4^2-x1\,x2^2\,x4^2-x0^2\,x3\,x4^2-$$
$$2\,x0\,x1\,x3\,x4^2-x1^2\,x3\,x4^2-2\,x0\,x2\,x3\,x4^2-2\,x1\,x2\,x3\,x4^2-x0\,x3^2\,x4^2-x2\,x3^2\,x4^2\}$$

Out[1153]=

$$\{0,\ \sqrt{5}\ (-x0\,x1+x0\,x2-x1\,x2+x0\,x3+x1\,x3-x2\,x3-x0\,x4+x1\,x4+x2\,x4-x3\,x4),$$
$$\sqrt{5}\ \big(x0^2\,x1+x0\,x1^2-x0^2\,x2+2\,x0\,x1\,x2+x1^2\,x2-x0\,x2^2+x1\,x2^2-x0^2\,x3-2\,x0\,x1\,x3-x1^2\,x3-$$
$$2\,x0\,x2\,x3+2\,x1\,x2\,x3+x2^2\,x3-x0\,x3^2-x1\,x3^2+x2\,x3^2+x0^2\,x4+2\,x0\,x1\,x4-x1^2\,x4-2\,x0\,x2\,x4-$$
$$2\,x1\,x2\,x4-x2^2\,x4+2\,x0\,x3\,x4-2\,x1\,x3\,x4+2\,x2\,x3\,x4+x3^2\,x4+x0\,x4^2-x1\,x4^2-x2\,x4^2+x3\,x4^2\big),$$
$$\sqrt{5}\ \big(-x0^2\,x1^2-x0^2\,x1\,x2-2\,x0\,x1^2\,x2+x0^2\,x2^2-x0\,x1\,x2^2-x1^2\,x2^2+x0^2\,x1\,x3+x0\,x1^2\,x3+$$
$$2\,x0^2\,x2\,x3-x1^2\,x2\,x3+x0\,x2^2\,x3-2\,x1\,x2^2\,x3+x0^2\,x3^2+2\,x0\,x1\,x3^2+x1^2\,x3^2+x0\,x2\,x3^2-$$
$$x1\,x2\,x3^2-x2^2\,x3^2-2\,x0^2\,x1\,x4-x0\,x1^2\,x4+x0^2\,x2\,x4+x1^2\,x2\,x4+2\,x0\,x2^2\,x4+x1\,x2^2\,x4-$$
$$x0^2\,x3\,x4+2\,x1^2\,x3\,x4-x2^2\,x3\,x4-x0\,x3^2\,x4+x1\,x3^2\,x4-2\,x2\,x3^2\,x4-x0^2\,x4^2-x0\,x1\,x4^2+$$
$$x1^2\,x4^2+x0\,x2\,x4^2+2\,x1\,x2\,x4^2+x2^2\,x4^2-2\,x0\,x3\,x4^2+x1\,x3\,x4^2-x2\,x3\,x4^2-x3^2\,x4^2\big),$$
$$\sqrt{5}\ \big(x0^2\,x1^2\,x2+x0\,x1^2\,x2^2-x0^2\,x2^2\,x3+x1^2\,x2^2\,x3-x0^2\,x1\,x3^2-x0\,x1^2\,x3^2-x0^2\,x2\,x3^2+$$
$$x1\,x2^2\,x3^2+x0^2\,x1^2\,x4-x0^2\,x2^2\,x4-x1^2\,x3^2\,x4+x2^2\,x3^2\,x4+x0^2\,x1\,x4^2-x1^2\,x2\,x4^2-$$
$$x0\,x2^2\,x4^2-x1\,x2^2\,x4^2+x0^2\,x3\,x4^2-x1^2\,x3\,x4^2+x0\,x3^2\,x4^2+x2\,x3^2\,x4^2\big)\}$$

【コメント】d_i, e_i の定義は「非常に」分かりにくいです．アルファベット順を考えると F^+, $\sqrt{5}\ F^-$ の係数をそれぞれ d_i, e_i とすべきと思うのですが，逆になっています．
ここでは原論文の通りにしました．それぐらいはまだ良いのですが，もっと問題なのは，
d_i の定義は 本文では「$\sqrt{D}\ (f_1-f_2)$ の $(5-i)$ 次の係数」となっていて，
この後の「解の公式」が導けず，苦労しました．
しかし章末にあるアルゴリズムを読むと「$\sqrt{5}\ (f_1-f_2)$ の $(5-i)$ 次の係数」となっているので，
定義を変更する事により上手くいきました．更には，一般の関数F_1,
F_2 に対しては，D_i は「(F_1-F_2) の $(5-i)$ 次の係数」となっています．
なんと3通りも異なる定義が使われています．(しかも本文のd_i, e_i の定義は上述の様に間違いです！)
また7次方程式でも F_1, F_2, F^+, F^- は定義されますが，その係数はそれぞれ b_i,
c_i, d^+, d^- となっていて5次方程式とは異なる文字が使われています．

§4. 解の公式A とその証明

以下「$f = x^5 + a_2 x^3 + a_3 x^2 + a_4 x + a_5$ $(a_1 = 0)$」と仮定します.

解の公式は a_i, d_i, e_i で表す事ができます.
（他にも色々な表現があります.なので解の公式「A」としてます.）

§4 – 0　解の公式A

ξ を1の虚数5乗根とするとき, Lagrange 分解式は次の$r_0 \sim r_4$ になります.
（原論文では Fourier transform と呼んでいます.）

In[1154]:=

```
Clear[x0, x1, x2, x3, x4]
r0 = x0 + x1 + x2 + x3 + x4 ;
r1 = x0 + ξ x1 + ξ^2 x2 + ξ^3 x3 + ξ^4 x4;
r2 = x0 + ξ^2 x1 + ξ^4 x2 + ξ x3 + ξ^3 x4;
r3 = x0 + ξ^3 x1 + ξ x2 + ξ^4 x3 + ξ^2 x4;
r4 = x0 + ξ^4 x1 + ξ^3 x2 + ξ^2 x3 + ξ x4;
```

σ, τ によって r_k は次の様に移ります.

「σ : $r_0 \to r_0$, $r_1 \to \xi^4 r_1$, $r_2 \to \xi^3 r_2$, $r_3 \to \xi^2 r_3$, $r_4 \to \xi r_4$」,
「τ : $r_k \to r_{2k}$, τ^2 : $r_k \to r_{4k}$」

そして解の公式 （の一歩手前） は次の4式となりま
す. まず両辺の固定群を調べておくと, 左辺は F_{10} で動かず F_{20} で動きます. 右辺も同様です.

$$\begin{cases} r_1 r_4 = -\frac{1}{2} d_2 - \frac{5}{2} a_2 & (1) \\ r_1^5 + r_4^5 = \frac{125}{2} d_5 + 125 e_5 - \frac{25}{4} d_3 a_2 - \frac{75}{4} d_2 a_3 - \frac{125}{2} a_2 a_3 - \frac{375}{2} a_5 & (2) \\ r_4^2 r_2 + r_1^2 r_3 = -\frac{5}{2} d_3 - \frac{25}{2} a_3 & (3) \\ r_1^3 r_2 + r_4^3 r_3 = \frac{25}{2} d_4 + \frac{15}{2} e_4 - \frac{15}{2} d_2 a_2 - 40 a_4 - \frac{5}{2} a_2^2 & (4) \end{cases}$$

この式を証明するには全てを $x_0 \sim x_4$ の式に直して証明します. r_i, e_i,
d_i は既に そうなっています. また a_i を $x_0 \sim x_4$ の式に直した式を $\$a_i$ とおき,
k次の基本対称式をs_k とおきます. 更にイデアル $I = \{s1\ ,\ s2 - a2,\ s3 + a3,\ s4 - a4,\ s5 + a5\}$
のグレブナー基底を作り base と置きます. （s_k の名前付け）

GroebnerBasis[{$poly_1$, $poly_2$, …}, {x_1, x_2, …}]
多項式の集合 $poly_i$についてグレブナー (Gröbner) 基底を形成する多項式を求める

GroebnerBasis [{$poly_1$, $poly_2$, …}, {x_1, x_2, …}, {y_1, y_2, …}]
変数y_i を消去したグレブナー基底を求める.

In[1160]:=

```
{s1, s2, s3, s4, s5} = Table[SymmetricPolynomial[k, {x0, x1, x2, x3, x4}], {k, 1, 5}]
{$a1, $a2, $a3, $a4, $a5} = {(-1), 1, (-1), 1, (-1)} {s1, s2, s3, s4, s5}
base = GroebnerBasis[{s1 , s2 - a2, s3 + a3, s4 - a4, s5 + a5}, {x4, x3, x2, x1, x0}]
```

Out[1160]=

```
{x0 + x1 + x2 + x3 + x4, x0 x1 + x0 x2 + x1 x2 + x0 x3 + x1 x3 + x2 x3 + x0 x4 + x1 x4 + x2 x4 + x3 x4,
 x0 x1 x2 + x0 x1 x3 + x0 x2 x3 + x1 x2 x3 + x0 x1 x4 + x0 x2 x4 + x1 x2 x4 + x0 x3 x4 + x1 x3 x4 +
  x2 x3 x4, x0 x1 x2 x3 + x0 x1 x2 x4 + x0 x1 x3 x4 + x0 x2 x3 x4 + x1 x2 x3 x4, x0 x1 x2 x3 x4}
```

Out[1161]=

```
{-x0 - x1 - x2 - x3 - x4, x0 x1 + x0 x2 + x1 x2 + x0 x3 + x1 x3 + x2 x3 + x0 x4 + x1 x4 + x2 x4 + x3 x4,
 -x0 x1 x2 - x0 x1 x3 - x0 x2 x3 - x1 x2 x3 - x0 x1 x4 - x0 x2 x4 - x1 x2 x4 - x0 x3 x4 - x1 x3 x4 -
  x2 x3 x4, x0 x1 x2 x3 + x0 x1 x2 x4 + x0 x1 x3 x4 + x0 x2 x3 x4 + x1 x2 x3 x4, -x0 x1 x2 x3 x4}
```

Out[1162]=

$\{a5 + a4\, x0 + a3\, x0^2 + a2\, x0^3 + x0^5,$
$a4 + a3\, x0 + a2\, x0^2 + x0^4 + a3\, x1 + a2\, x0\, x1 + x0^3\, x1 + a2\, x1^2 + x0^2\, x1^2 + x0\, x1^3 + x1^4,\ a3 + a2\, x0 +$
$x0^3 + a2\, x1 + x0^2\, x1 + x0\, x1^2 + x1^3 + a2\, x2 + x0^2\, x2 + x0\, x1\, x2 + x1^2\, x2 + x0\, x2^2 + x1\, x2^2 + x2^3,$
$a2 + x0^2 + x0\, x1 + x1^2 + x0\, x2 + x1\, x2 + x2^2 + x0\, x3 + x1\, x3 + x2\, x3 + x3^2,\ x0 + x1 + x2 + x3 + x4\}$

【コメント】「base」は 基本対称式 s_kに $(-1)^k a_k$ という名前を付けただけです. (k=2,..,5) 例えば次と同じ使い方です.

「問」$x_1^3 + x_2^3 + x_3^3$ を $x_1 + x_2 + x_3$, $x_1 x_2 + x_2 x_3 + x_3 x_1$, $x_1 x_2 x_3$ を用いて表せ

「解答」
```
grb = GroebnerBasis[
        {x1 + x2 + x3 - p, x1 x2 + x2 x3 + x3 x1 - q, x1 x2 x3 - r}, {x3, x2, x1, p, q, r}]
```
$\{-r + q\, x1 - p\, x1^2 + x1^3,\ q - p\, x1 + x1^2 - p\, x2 + x1\, x2 + x2^2,\ -p + x1 + x2 + x3\}$
```
PolynomialReduce[x1^3 + x2^3 + x3^3, grb, {x3, x2, x1, p, q, r}]
```
$\{\{3,\ 2p - 2\, x1 + x2 + x3,\ p^2 - q - p\, x1 - p\, x2 + x1\, x2 + p\, x3 - x1\, x3 - x2\, x3 + x3^2\},$
$p^3 - 3pq + 3r\}$

第2要素は「$x_1^3 + x_2^3 + x_3^3 =$
$(x_1 + x_2 + x_3)^3 - 3(x_1 + x_2 + x_3)(x_1 x_2 + x_2 x_3 + x_3 x_1) + 3\ x_1 x_2 x_3$」を表します.

(1)~(4) の導き方は原論文の著者によると次の様になります. ちなみに「G = base」です.

For the first invariant F1, we compute its normal NF (F1) form by G.If it belongs to a basis of the free module of invariants (i.e.its leading term depends only on the x_i), we introduce a new indeterminate f1 and the polynomial P1 := NF (F1) - f1.From now on, the normal form procedure is modified and consists in reducing first by G, then by P1, . . ., a polynomial being reducible by some Pi only if its leading term is the product of the leading term of Pi by a monomial which is independent of the x_i.

上の方針に従うと, 次の様になると思います. (残念ながら原論文では式の形では述べられていないので確信はありません.)

なお, Mathematica では「簡約」は, PolynomialReduce (その内でも, 余り=第2成分の方) を使います. 「整式のMod」は PolynomialModを使います. PolynomialReduce と PolynomialMod はほぼ同じ

ですが，私は複数の整式（特にグレブナー基底）で割るときは PolynomialReduce を，一個の式で割るときは PolynomialMod を使っています．更に詳しいことは Google を御参照ください．

PolynomialReduce[$poly$, {$poly_1$, $poly_2$, …}, {x_1, x_2, …}]

$poly_i$ によって $poly$ を簡約したリストを返す．求まるリストは{{a_1, a_2, …}, b}の形であり，b は最小で，$poly$ は $a_1\ poly_1 + a_2\ poly_2 + \ldots + b$ に等しい．

PolynomialMod[$poly$, m]

m を法として多項式 $poly$ を与える．

PolynomialMod[$poly$, {m_1, m_2, …}]

すべての m_iを法として簡約する．

§4 − 1 $r_1 r_4$ について

まずは poly1 = $r_1 r_4$ を x_0 ~ x_4 の式で表して

「$1+\zeta+\zeta^2+\zeta^3+\zeta^4=0$」と「$\zeta^2+\zeta^3=\left(-1-\sqrt{5}\right)/2$」を使って簡単にします.

In[1163]:=

```
poly1 = r1 * r4 // PolynomialMod[#, Cyclotomic[5, ξ]] & // Collect[#, ξ] &
poly1 = poly1 /. {ξ^3 → 2 Cos[4 / 5 Pi] - ξ^2} // FullSimplify
```

Out[1163]=

$$x0^2 - x0\,x1 + x1^2 - x1\,x2 + x2^2 - x2\,x3 + x3^2 - x0\,x4 - x3\,x4 + x4^2 + (-x0\,x1 + x0\,x2 - x1\,x2 + x0\,x3 + x1\,x3 - x2\,x3 - x0\,x4 + x1\,x4 + x2\,x4 - x3\,x4)\,\xi^2 + (-x0\,x1 + x0\,x2 - x1\,x2 + x0\,x3 + x1\,x3 - x2\,x3 - x0\,x4 + x1\,x4 + x2\,x4 - x3\,x4)\,\xi^3$$

Out[1164]=

$$\frac{1}{2}\left(2\,x0^2 + \left(-1+\sqrt{5}\right) x0\,x1 + 2\,x1^2 + \left(-1+\sqrt{5}\right) x1\,x2 + 2\,x2^2 + 2\,x3^2 - x2\,x4 - \sqrt{5}\,x2\,x4 + 2\,x4^2 + \left(-1+\sqrt{5}\right) x3\,(x2+x4) - \left(1+\sqrt{5}\right) x1\,(x3+x4) - x0\left(x2+\sqrt{5}\,x2 + x3 + \sqrt{5}\,x3 + x4 - \sqrt{5}\,x4\right)\right)$$

poly1, d2 を base で簡約した余りを求め，それぞれ先頭に$を付けた名前を付けます.

In[1165]:=

```
$d2 = PolynomialReduce[d2, base, {x4, x3, x2, x1, x0}] [[2]]
$poly1 = PolynomialReduce[poly1, base, {x4, x3, x2, x1, x0}] [[2]]
```

Out[1165]=

$$-\sqrt{5}\,a2 - 2\sqrt{5}\,x0\,x1 - 2\sqrt{5}\,x1^2 - 4\sqrt{5}\,x1\,x2 - 2\sqrt{5}\,x2^2 + 2\sqrt{5}\,x0\,x3 - 2\sqrt{5}\,x2\,x3$$

Out[1166]=

$$\frac{1}{2}\left(-5\,a2 + \sqrt{5}\,a2 + 2\sqrt{5}\,x0\,x1 + 2\sqrt{5}\,x1^2 + 4\sqrt{5}\,x1\,x2 + 2\sqrt{5}\,x2^2 - 2\sqrt{5}\,x0\,x3 + 2\sqrt{5}\,x2\,x3\right)$$

$d2のLT（Leading Term）は x_i の式なので，$d2は独立変数です.
よって　$poly1を（$d2 − m1）で簡約します（m1は$d2の名前で 不定変数（indeterminate））

In[1167]:=

```
PolynomialReduce[$poly1, $d2 - m1, {x4, x3, x2, x1, x0}]
```

Out[1167]=

$$\left\{\left\{-\frac{1}{2}\right\}, \frac{1}{2}\,(-5\,a2 - m1)\right\}$$

これは「**poly1 = $-\frac{1}{2}$ ($d2-m1)+$\frac{1}{2}$(-5 a2 - m1)**」を表しますが，m1は$d2の名前なので，($d2-m1)は零ベクトルです．故に「poly1 = (簡約した余り) = $\frac{1}{2}$ (−5 a2 − m1)」となります.

【検算】

私はグレブナー基底を使った計算はやや不安なので，検算します

In[1168]:=

```
Factor[poly1 + 1 / 2 d2 + 5 / 2 $a2]
```

Out[1168]=

$$(x0 + x1 + x2 + x3 + x4)^2$$

「$x_0 + x_1 + x_2 + x_3 + x_4 = 0$」なので，確かに式（1）は正しいです.

§4 − 2 $r_1{}^5 + r_4{}^5$ について

まずは poly2 = $r_1{}^5 + r_4{}^5$ を $x_0 \sim x_4$ の式で表して

「$1 + \zeta + \zeta^2 + \zeta^3 + \zeta^4 = 0$」と「$\zeta^2 + \zeta^3 = \left(-1 - \sqrt{5}\right)/2$」を使って簡単にします.

In[1169]:=

```
poly2 = (r1^5 + r4^5 // PolynomialMod[#, Cyclotomic[5, ξ]] & // Collect[#, ξ] &) /.
  {ξ^2 → 2 Cos[4 / 5 Pi] - ξ^3} // Simplify
```

Out[1169]=

$$
\begin{aligned}
&\frac{1}{2}\Big(4\,x0^{5} + 4\,x1^{5} + 4\,x2^{5} - 5\,x2^{4}\,x3 + 5\sqrt{5}\,x2^{4}\,x3 - 10\,x2^{3}\,x3^{2} - 10\sqrt{5}\,x2^{3}\,x3^{2} - 10\,x2^{2}\,x3^{3} - 10\sqrt{5}\,x2^{2}\,x3^{3} -\\
&\quad 5\,x2\,x3^{4} + 5\sqrt{5}\,x2\,x3^{4} + 4\,x3^{5} - 5\,x2^{4}\,x4 - 5\sqrt{5}\,x2^{4}\,x4 - 20\,x2^{3}\,x3\,x4 - 20\sqrt{5}\,x2^{3}\,x3\,x4 - 30\,x2^{2}\,x3^{2}\,x4 +\\
&\quad 30\sqrt{5}\,x2^{2}\,x3^{2}\,x4 + 80\,x2\,x3^{3}\,x4 - 5\,x3^{4}\,x4 + 5\sqrt{5}\,x3^{4}\,x4 - 10\,x2^{3}\,x4^{2} + 10\sqrt{5}\,x2^{3}\,x4^{2} + 120\,x2^{2}\,x3\,x4^{2} -\\
&\quad 30\,x2\,x3^{2}\,x4^{2} + 30\sqrt{5}\,x2\,x3^{2}\,x4^{2} - 10\,x3^{3}\,x4^{2} - 10\sqrt{5}\,x3^{3}\,x4^{2} - 10\,x2^{2}\,x4^{3} + 10\sqrt{5}\,x2^{2}\,x4^{3} - 20\,x2\,x3\,x4^{3} -\\
&\quad 20\sqrt{5}\,x2\,x3\,x4^{3} - 10\,x3^{2}\,x4^{3} - 10\sqrt{5}\,x3^{2}\,x4^{3} - 5\,x2\,x4^{4} - 5\sqrt{5}\,x2\,x4^{4} - 5\,x3\,x4^{4} + 5\sqrt{5}\,x3\,x4^{4} + 4\,x4^{5} +\\
&\quad 5\,x0^{4}\left(\left(-1+\sqrt{5}\right)x1 - \left(1+\sqrt{5}\right)x2 - x3 - \sqrt{5}\,x3 - x4 + \sqrt{5}\,x4\right) + 5\,x1^{4}\left(\left(-1+\sqrt{5}\right)x2 - \left(1+\sqrt{5}\right)(x3+x4)\right) -\\
&\quad 10\,x1^{3}\left(\left(1+\sqrt{5}\right)x2^{2} - \left(-1+\sqrt{5}\right)x3^{2} - 8\,x3\,x4 - \left(-1+\sqrt{5}\right)x4^{2} + 2\,x2\left(x3 + \sqrt{5}\,x3 + x4 - \sqrt{5}\,x4\right)\right) -\\
&\quad 10\,x0^{3}\Big(\left(1+\sqrt{5}\right)x1^{2} - \left(-1+\sqrt{5}\right)x2^{2} + x3^{2} - \sqrt{5}\,x3^{2} + 2\,x1\left(x2 + \sqrt{5}\,x2 + x3 - \sqrt{5}\,x3 - 4\,x4\right) +\\
&\qquad 2\,x3\,x4 + 2\sqrt{5}\,x3\,x4 + x4^{2} + \sqrt{5}\,x4^{2} - 2\,x2\left(4\,x3 + \left(-1+\sqrt{5}\right)x4\right)\Big) +\\
&\quad 5\,x1\Big(\left(-1+\sqrt{5}\right)x2^{4} - \left(1+\sqrt{5}\right)x3^{4} + 4\left(-1+\sqrt{5}\right)x3^{3}\,x4 + 24\,x3^{2}\,x4^{2} + 4\left(-1+\sqrt{5}\right)x3\,x4^{3} - \left(1+\sqrt{5}\right)x4^{4} +\\
&\qquad 4\,x2^{3}\left(4\,x3 + \left(-1+\sqrt{5}\right)x4\right) + 6\,x2^{2}\left(\left(-1+\sqrt{5}\right)x3^{2} - 2\left(1+\sqrt{5}\right)x3\,x4 - \left(1+\sqrt{5}\right)x4^{2}\right) -\\
&\qquad 4\,x2\left(\left(1+\sqrt{5}\right)x3^{3} + 3\left(1+\sqrt{5}\right)x3^{2}\,x4 - 3\left(-1+\sqrt{5}\right)x3\,x4^{2} - 4\,x4^{3}\right)\Big) -\\
&\quad 10\,x0^{2}\Big(\left(1+\sqrt{5}\right)x1^{3} - \left(-1+\sqrt{5}\right)x2^{3} + x3^{3} - \sqrt{5}\,x3^{3} - 12\,x3^{2}\,x4 + 3\,x3\,x4^{2} - 3\sqrt{5}\,x3\,x4^{2} + x4^{3} + \sqrt{5}\,x4^{3} + 3\left(1+\sqrt{5}\right)\\
&\qquad\ x2^{2}(x3+x4) - 3\,x1^{2}\left(\left(-1+\sqrt{5}\right)x2 + 4\,x3 + \left(-1+\sqrt{5}\right)x4\right) + 3\,x2\left(\left(1+\sqrt{5}\right)x3^{2} - 2\left(-1+\sqrt{5}\right)x3\,x4 - 4\,x4^{2}\right) +\\
&\qquad 3\,x1\left(-4\,x2^{2} + \left(1+\sqrt{5}\right)x3^{2} + 2\left(1+\sqrt{5}\right)x3\,x4 - \left(-1+\sqrt{5}\right)x4^{2} + 2\,x2\left(x3 - \sqrt{5}\,x3 + x4 + \sqrt{5}\,x4\right)\right)\Big) -\\
&\quad 10\,x1^{2}\Big(\left(1+\sqrt{5}\right)x2^{3} - 3\,x2^{2}\left(\left(-1+\sqrt{5}\right)x3 + 4\,x4\right) + 3\,x2\left(-4\,x3^{2} - 2\left(-1+\sqrt{5}\right)x3\,x4 + \left(1+\sqrt{5}\right)x4^{2}\right) -\\
&\qquad (x3+x4)\left(\left(-1+\sqrt{5}\right)x3^{2} + \left(-1+\sqrt{5}\right)x4^{2} - 2\,x3\left(x4 + 2\sqrt{5}\,x4\right)\right)\Big) +\\
&\quad 5\,x0\Big(\left(-1+\sqrt{5}\right)x1^{4} - \left(1+\sqrt{5}\right)x2^{4} - x3^{4} - \sqrt{5}\,x3^{4} - 4\,x3^{3}\,x4 - 4\sqrt{5}\,x3^{3}\,x4 - 6\,x3^{2}\,x4^{2} +\\
&\qquad 6\sqrt{5}\,x3^{2}\,x4^{2} + 16\,x3\,x4^{3} - x4^{4} + \sqrt{5}\,x4^{4} + 4\,x2^{3}\left(\left(-1+\sqrt{5}\right)x3 + 4\,x4\right) +\\
&\qquad 4\,x1^{3}\left(4\,x2 + \left(-1+\sqrt{5}\right)x3 - \left(1+\sqrt{5}\right)x4\right) + 6\,x2^{2}\left(4\,x3^{2} + 2\left(-1+\sqrt{5}\right)x3\,x4 - \left(1+\sqrt{5}\right)x4^{2}\right) +\\
&\qquad 6\,x1^{2}\left(\left(-1+\sqrt{5}\right)x2^{2} - \left(1+\sqrt{5}\right)x3^{2} + 2\left(-1+\sqrt{5}\right)x3\,x4 + 4\,x4^{2} - 2\left(1+\sqrt{5}\right)x2\,(x3+x4)\right) +\\
&\qquad 4\,x2\,(x3+x4)\left(\left(-1+\sqrt{5}\right)x3^{2} + \left(-1+\sqrt{5}\right)x4^{2} - 2\,x3\left(x4 + 2\sqrt{5}\,x4\right)\right) -\\
&\qquad 4\,x1\Big(\left(1+\sqrt{5}\right)x2^{3} - 4\,x3^{3} - 3\left(-1+\sqrt{5}\right)x3^{2}\,x4 + 3\left(1+\sqrt{5}\right)x3\,x4^{2} + \left(1+\sqrt{5}\right)x4^{3} +\\
&\qquad\quad 3\,x2^{2}\left(x3 + \sqrt{5}\,x3 + x4 - \sqrt{5}\,x4\right) - 3\,x2\left(\left(-1+\sqrt{5}\right)x3^{2} + 8\,x3\,x4 + \left(-1+\sqrt{5}\right)x4^{2}\right)\Big)\Big)\Big)
\end{aligned}
$$

次に poly2, d5, e5, d3, d2 を base で簡約した余りを求め，それぞれ先頭に$を付けた名を与えます.

In[1170]:=

```
{$poly2, $d5, $e5, $d3, $d2 } =
 PolynomialReduce[#, base, {x4, x3, x2, x1, x0}][[2]] & /@ {poly2, d5, e5, d3 , d2 }
```

Out[1170]=

$$
\begin{aligned}
&\Big\{-\frac{25}{2}\\
&\quad\big(-5\,a2\,a3 + 3\sqrt{5}\,a2\,a3 + 25\,a5 - 25\sqrt{5}\,a5 - 10\sqrt{5}\,a4\,x0 - 10\,a2^2\,x1 + 4\sqrt{5}\,a2^2\,x1 + 20\,a4\,x1 + 30\,a3\,x0\,x1 + 12\sqrt{5}\,a3\,x0\,x1 +\\
&\quad 10\,a2\,x0^2\,x1 + 24\sqrt{5}\,a2\,x0^2\,x1 + 20\,x0^4\,x1 + 20\sqrt{5}\,x0^4\,x1 + 10\,a3\,x1^2 + 2\sqrt{5}\,a3\,x1^2 + 10\,a2\,x0\,x1^2 + 4\sqrt{5}\,a2\,x0\,x1^2 +\\
&\quad 10\sqrt{5}\,x0^3\,x1^2 - 10\,a2\,x1^3 + 4\sqrt{5}\,a2\,x1^3 - 20\,x0^2\,x1^3 + 10\sqrt{5}\,x0^2\,x1^3 + 20\,a4\,x2 + 10\sqrt{5}\,a4\,x2 + 20\sqrt{5}\,a3\,x0\,x2 +\\
&\quad 10\,a2\,x0^2\,x2 + 16\sqrt{5}\,a2\,x0^2\,x2 + 20\sqrt{5}\,x0^4\,x2 + 20\,a3\,x1\,x2 + 4\sqrt{5}\,a3\,x1\,x2 - 10\,a2\,x0\,x1\,x2 + 6\sqrt{5}\,a2\,x0\,x1\,x2 -\\
&\quad 20\,x0^3\,x1\,x2 - 40\,x0^2\,x1^2\,x2 - 20\sqrt{5}\,x0^2\,x1^2\,x2 - 40\,x0\,x1^3\,x2 - 10\sqrt{5}\,x0\,x1^3\,x2 - 10\,a3\,x2^2 + 2\sqrt{5}\,a3\,x2^2 - 10\,a2\,x0\,x2^2 -\\
&\quad 4\sqrt{5}\,a2\,x0\,x2^2 - 20\,x0^3\,x2^2 - 20\,a2\,x1\,x2^2 - 40\,x0^2\,x1\,x2^2 - 10\sqrt{5}\,x0^2\,x1\,x2^2 - 20\,x0\,x1^2\,x2^2 - 20\sqrt{5}\,x0\,x1^2\,x2^2 -\\
&\quad 20\,x1^3\,x2^2 + 10\sqrt{5}\,a4\,x3 - 10\,a3\,x0\,x3 + 8\sqrt{5}\,a3\,x0\,x3 - 20\,a2\,x0^2\,x3 + 10\sqrt{5}\,a2\,x0^2\,x3 - 20\,x0^4\,x3 + 10\sqrt{5}\,x0^4\,x3 -\\
&\quad 10\,a2\,x0\,x1\,x3 - 6\sqrt{5}\,a2\,x0\,x1\,x3 - 20\,x0^3\,x1\,x3 - 10\sqrt{5}\,x0^3\,x1\,x3 - 40\,x0^2\,x1^2\,x3 - 10\sqrt{5}\,x0^2\,x1^2\,x3 - 20\,x0\,x1^3\,x3 -\\
&\quad 10\sqrt{5}\,x0\,x1^3\,x3 - 10\,a3\,x2\,x3 + 2\sqrt{5}\,a3\,x2\,x3 - 20\,a2\,x0\,x2\,x3 - 40\,x0^3\,x2\,x3 - 10\,a2\,x1\,x2\,x3 - 4\sqrt{5}\,a2\,x1\,x2\,x3 -\\
&\quad 40\,x0^2\,x1\,x2\,x3 - 20\sqrt{5}\,x0^2\,x1\,x2\,x3 - 20\,x0\,x1^2\,x2\,x3 - 10\sqrt{5}\,x0\,x1^2\,x2\,x3 - 20\,x0^2\,x2^2\,x3 + 20\,x1^2\,x2^2\,x3\big),\\
&-\sqrt{5}\,a2\,a3 + 5\sqrt{5}\,a5 + 2\sqrt{5}\,a4\,x0 - \sqrt{5}\,a2^2\,x1 - 3\sqrt{5}\,a3\,x0\,x1 - 5\sqrt{5}\,a2\,x0^2\,x1 - 4\sqrt{5}\,x0^4\,x1 -\\
&\sqrt{5}\,a3\,x1^2 - \sqrt{5}\,a2\,x0\,x1^2 - 2\sqrt{5}\,x0^3\,x1^2 - \sqrt{5}\,a2\,x1^3 - 2\sqrt{5}\,x0^2\,x1^3 -\\
&2\sqrt{5}\,a4\,x2 - 4\sqrt{5}\,a3\,x0\,x2 - 3\sqrt{5}\,a2\,x0^2\,x2 - 4\sqrt{5}\,x0^4\,x2 - 2\sqrt{5}\,a3\,x1\,x2 -\\
&\sqrt{5}\,a2\,x0\,x1\,x2 + 4\sqrt{5}\,x0^2\,x1^2\,x2 + 2\sqrt{5}\,x0\,x1^3\,x2 - \sqrt{5}\,a3\,x2^2 + \sqrt{5}\,a2\,x0\,x2^2 +\\
&2\sqrt{5}\,x0^2\,x1\,x2^2 + 4\sqrt{5}\,x0\,x1^2\,x2^2 - 2\sqrt{5}\,a4\,x3 - \sqrt{5}\,a3\,x0\,x3 - 2\sqrt{5}\,a2\,x0^2\,x3 -\\
&2\sqrt{5}\,x0^4\,x3 + \sqrt{5}\,a2\,x0\,x1\,x3 + 2\sqrt{5}\,x0^3\,x1\,x3 + 2\sqrt{5}\,x0^2\,x1^2\,x3 + 2\sqrt{5}\,x0\,x1^3\,x3 -\\
&\sqrt{5}\,a3\,x2\,x3 + \sqrt{5}\,a2\,x1\,x2\,x3 + 4\sqrt{5}\,x0^2\,x1\,x2\,x3 + 2\sqrt{5}\,x0\,x1^2\,x2\,x3,\\
&a2\,a3 - a5 + \left(a2^2 - 2\,a4\right)\,x1 - 3\,a3\,x0\,x1 - a2\,x0^2\,x1 - 2\,x0^4\,x1 - a3\,x1^2 - a2\,x0\,x1^2 + a2\,x1^3 + 2\,x0^2\,x1^3 -\\
&2\,a4\,x2 - a2\,x0^2\,x2 - 2\,a3\,x1\,x2 + a2\,x0\,x1\,x2 + 2\,x0^3\,x1\,x2 + 4\,x0^2\,x1^2\,x2 + 4\,x0\,x1^3\,x2 + a3\,x2^2 +\\
&a2\,x0\,x2^2 + 2\,x0^3\,x2^2 + 2\,a2\,x1\,x2^2 + 4\,x0^2\,x1\,x2^2 + 2\,x0\,x1^2\,x2^2 + 2\,x1^3\,x2^2 + a3\,x0\,x3 + 2\,a2\,x0^2\,x3 +\\
&2\,x0^4\,x3 + a2\,x0\,x1\,x3 + 2\,x0^3\,x1\,x3 + 4\,x0^2\,x1^2\,x3 + 2\,x0\,x1^3\,x3 + a3\,x2\,x3 + 2\,a2\,x0\,x2\,x3 +\\
&4\,x0^3\,x2\,x3 + a2\,x1\,x2\,x3 + 4\,x0^2\,x1\,x2\,x3 + 2\,x0\,x1^2\,x2\,x3 + 2\,x0^2\,x2^2\,x3 - 2\,x1^2\,x2^2\,x3,\\
&-\sqrt{5}\,a3 - 2\sqrt{5}\,a2\,x1 - 2\sqrt{5}\,x0^2\,x1 - 2\sqrt{5}\,x0\,x1^2 - 2\sqrt{5}\,x1^3 + 2\sqrt{5}\,x0^2\,x2 +\\
&2\sqrt{5}\,x0\,x1\,x2 + 2\sqrt{5}\,x0\,x2^2 - 2\sqrt{5}\,x0\,x1\,x3 + 2\sqrt{5}\,x1\,x2\,x3,\\
&-\sqrt{5}\,a2 - 2\sqrt{5}\,x0\,x1 - 2\sqrt{5}\,x1^2 - 4\sqrt{5}\,x1\,x2 - 2\sqrt{5}\,x2^2 + 2\sqrt{5}\,x0\,x3 - 2\sqrt{5}\,x2\,x3\Big\}
\end{aligned}
$$

まず$d5のLTは x_i の式なので, $d5は独立変数です． よって 次は $e5を($d5-m1)で簡約して $$e5という名前を付けます．(m1は$d5の名前で 不定変数(indeterminate))

In[1171]:=

```
$$e5 = PolynomialReduce[$e5, $d5 - m1, {x4, x3, x2, x1, x0}][[2]]
```

Out[1171]=

$$
\begin{aligned}
&\frac{2\sqrt{5}\,a2\,a3 - 6\sqrt{5}\,a5 + m1}{\sqrt{5}} - 2\,a4\,x0 + 2\left(a2^2 - a4\right)\,x1 + 4\,a2\,x0^2\,x1 + 2\,x0^4\,x1 + 2\,x0^3\,x1^2 + 2\,a2\,x1^3 +\\
&4\,x0^2\,x1^3 + 4\,a3\,x0\,x2 + 2\,a2\,x0^2\,x2 + 4\,x0^4\,x2 + 2\,a2\,x0\,x1\,x2 + 2\,x0^3\,x1\,x2 + 2\,x0\,x1^3\,x2 + 2\,a3\,x2^2 +\\
&2\,x0^3\,x2^2 + 2\,a2\,x1\,x2^2 + 2\,x0^2\,x1\,x2^2 - 2\,x0\,x1^2\,x2^2 + 2\,x1^3\,x2^2 + 2\,a4\,x3 + 2\,a3\,x0\,x3 + 4\,a2\,x0^2\,x3 +\\
&4\,x0^4\,x3 + 2\,x0^2\,x1^2\,x3 + 2\,a3\,x2\,x3 + 2\,a2\,x0\,x2\,x3 + 4\,x0^3\,x2\,x3 + 2\,x0^2\,x2^2\,x3 - 2\,x1^2\,x2^2\,x3
\end{aligned}
$$

$$e5のLTは x_i の式なので, 独立変数です． よって d3 a2 を{$b3 - m1, $$e5 - m2}で簡約して $$d3a2 という名前を付けます．(m2は$$e5の名前で 不定変数)．

In[1172]:=

```
$$d3a2 = PolynomialReduce[$d3 a2 , {$d5 - m1, $$e5 - m2}, {x4, x3, x2, x1, x0}][[2]]
```

Out[1172]=

$$
\begin{aligned}
&-\sqrt{5}\,a2\,a3 - 2\sqrt{5}\,a2^2\,x1 - 2\sqrt{5}\,a2\,x0^2\,x1 - 2\sqrt{5}\,a2\,x0\,x1^2 - 2\sqrt{5}\,a2\,x1^3 +\\
&2\sqrt{5}\,a2\,x0^2\,x2 + 2\sqrt{5}\,a2\,x0\,x1\,x2 + 2\sqrt{5}\,a2\,x0\,x2^2 - 2\sqrt{5}\,a2\,x0\,x1\,x3 + 2\sqrt{5}\,a2\,x1\,x2\,x3
\end{aligned}
$$

$$d3a2のLTは x_i の式なので, 独立変数です． よって $d2 a3 を{$b3 - m1, $$e5 - m2,$$d3a2-m3}で簡約します．(m3は$$d3a2の名前で 不定変数)．

In[1173]:=

```
$$d2a3 =
 PolynomialReduce[$d2 a3, {$d5 - m1, $$e5 - m2, $$d3a2 - m3}, {x4, x3, x2, x1, x0}][[2]]
```

Out[1173]=

$$-\sqrt{5}\ a2\ a3 - 2\sqrt{5}\ a3\ x0\ x1 - 2\sqrt{5}\ a3\ x1^2 - 4\sqrt{5}\ a3\ x1\ x2 - 2\sqrt{5}\ a3\ x2^2 + 2\sqrt{5}\ a3\ x0\ x3 - 2\sqrt{5}\ a3\ x2\ x3$$

$$d2a3のLTは x_i の式なので, 独立変数です. よって最後に poly2 を {$b3 - m1, $$e5 - m2, $$d3a2- m3,$$d2a3 - m4}で簡約します . (m4は$$d2a3 の名前で 不定変数) .

In[1174]:=

```
PolynomialReduce[$poly2,
   {$d5 - m1, $$e5 - m2, $$d3a2 - m3, $$d2a3 - m4 }, {x4, x3, x2, x1, x0}][[2]] // Expand
```

Out[1174]=

$$-\frac{125\ a2\ a3}{2} - \frac{375\ a5}{2} + \frac{125\ m1}{2} + 125\ m2 - \frac{25\ m3}{4} - \frac{75\ m4}{4}$$

これは「$r1^5 + r_4{}^5 = -\frac{125}{2}\ a2\ a3 - \frac{375}{2}\ a5 + 125\ e5 + \frac{125}{2}\ d5 - \frac{25}{4}\ d3\ a2 - \frac{75}{4}\ d2\ a3$」

　を意味します.

【検算】

In[1175]:=

```
poly2 - (125 / 2 d5  + 125 e5 - 25 / 4 d3 ($a2) -
    75 / 4 d2 ($a3) - 125 / 2 (s2) ($a3) - 375 / 2 ($a5)) // Factor
```

Out[1175]=

$$\begin{aligned}
&\frac{1}{4}(x0 + x1 + x2 + x3 + x4)\\
&\big(8\,x0^4 - 18\,x0^3\,x1 + 10\sqrt{5}\,x0^3\,x1 - 2\,x0^2\,x1^2 - 5\sqrt{5}\,x0^2\,x1^2 - 18\,x0\,x1^3 + 10\sqrt{5}\,x0\,x1^3 + 8\,x1^4 - 18\,x0^3\,x2 - 10\sqrt{5}\,x0^3\,x2 -\\
&4\,x0^2\,x1\,x2 - 40\sqrt{5}\,x0^2\,x1\,x2 + 196\,x0\,x1^2\,x2 + 30\sqrt{5}\,x0\,x1^2\,x2 - 18\,x1^3\,x2 + 10\sqrt{5}\,x1^3\,x2 - 2\,x0^2\,x2^2 + 5\sqrt{5}\,x0^2\,x2^2 -\\
&4\,x0\,x1\,x2^2 - 40\sqrt{5}\,x0\,x1\,x2^2 - 2\,x1^2\,x2^2 - 5\sqrt{5}\,x1^2\,x2^2 - 18\,x0\,x2^3 - 10\sqrt{5}\,x0\,x2^3 - 18\,x1\,x2^3 + 10\sqrt{5}\,x1\,x2^3 + 8\,x2^4 -\\
&18\,x0^3\,x3 - 10\sqrt{5}\,x0^3\,x3 - 4\,x0^2\,x1\,x3 + 40\sqrt{5}\,x0^2\,x1\,x3 - 4\,x0\,x1^2\,x3 + 40\sqrt{5}\,x0\,x1^2\,x3 - 18\,x1^3\,x3 - 10\sqrt{5}\,x1^3\,x3 +\\
&196\,x0^2\,x2\,x3 - 30\sqrt{5}\,x0^2\,x2\,x3 - 58\,x0\,x1\,x2\,x3 - 4\,x1^2\,x2\,x3 - 40\sqrt{5}\,x1^2\,x2\,x3 - 4\,x0\,x2^2\,x3 + 40\sqrt{5}\,x0\,x2^2\,x3 +\\
&196\,x1\,x2^2\,x3 + 30\sqrt{5}\,x1\,x2^2\,x3 - 18\,x2^3\,x3 + 10\sqrt{5}\,x2^3\,x3 - 2\,x0^2\,x3^2 + 5\sqrt{5}\,x0^2\,x3^2 + 196\,x0\,x1\,x3^2 - 30\sqrt{5}\,x0\,x1\,x3^2 -\\
&2\,x1^2\,x3^2 + 5\sqrt{5}\,x1^2\,x3^2 - 4\,x0\,x2\,x3^2 + 40\sqrt{5}\,x0\,x2\,x3^2 - 4\,x1\,x2\,x3^2 - 40\sqrt{5}\,x1\,x2\,x3^2 - 2\,x2^2\,x3^2 - 5\sqrt{5}\,x2^2\,x3^2 -\\
&18\,x0\,x3^3 - 10\sqrt{5}\,x0\,x3^3 - 18\,x1\,x3^3 - 10\sqrt{5}\,x1\,x3^3 - 18\,x2\,x3^3 + 10\sqrt{5}\,x2\,x3^3 + 8\,x3^4 - 18\,x0^3\,x4 + 10\sqrt{5}\,x0^3\,x4 +\\
&196\,x0^2\,x1\,x4 + 30\sqrt{5}\,x0^2\,x1\,x4 - 4\,x0\,x1^2\,x4 - 40\sqrt{5}\,x0\,x1^2\,x4 - 18\,x1^3\,x4 - 10\sqrt{5}\,x1^3\,x4 - 4\,x0^2\,x2\,x4 + 40\sqrt{5}\,x0^2\,x2\,x4 -\\
&58\,x0\,x1\,x2\,x4 - 4\,x1^2\,x2\,x4 + 40\sqrt{5}\,x1^2\,x2\,x4 + 196\,x0\,x2^2\,x4 - 30\sqrt{5}\,x0\,x2^2\,x4 - 4\,x1\,x2^2\,x4 + 40\sqrt{5}\,x1\,x2^2\,x4 - 18\,x2^3\,x4 -\\
&10\sqrt{5}\,x2^3\,x4 - 4\,x0^2\,x3\,x4 - 40\sqrt{5}\,x0^2\,x3\,x4 - 58\,x0\,x1\,x3\,x4 + 196\,x1^2\,x3\,x4 - 30\sqrt{5}\,x1^2\,x3\,x4 - 58\,x0\,x2\,x3\,x4 -\\
&58\,x1\,x2\,x3\,x4 - 4\,x2^2\,x3\,x4 - 40\sqrt{5}\,x2^2\,x3\,x4 - 4\,x0\,x3^2\,x4 - 40\sqrt{5}\,x0\,x3^2\,x4 - 4\,x1\,x3^2\,x4 + 40\sqrt{5}\,x1\,x3^2\,x4 +\\
&196\,x2\,x3^2\,x4 + 30\sqrt{5}\,x2\,x3^2\,x4 - 18\,x3^3\,x4 + 10\sqrt{5}\,x3^3\,x4 - 2\,x0^2\,x4^2 - 5\sqrt{5}\,x0^2\,x4^2 - 4\,x0\,x1\,x4^2 - 40\sqrt{5}\,x0\,x1\,x4^2 -\\
&2\,x1^2\,x4^2 + 5\sqrt{5}\,x1^2\,x4^2 - 4\,x0\,x2\,x4^2 + 40\sqrt{5}\,x0\,x2\,x4^2 + 196\,x1\,x2\,x4^2 - 30\sqrt{5}\,x1\,x2\,x4^2 - 2\,x2^2\,x4^2 + 5\sqrt{5}\,x2^2\,x4^2 +\\
&196\,x0\,x3\,x4^2 + 30\sqrt{5}\,x0\,x3\,x4^2 - 4\,x1\,x3\,x4^2 + 40\sqrt{5}\,x1\,x3\,x4^2 - 4\,x2\,x3\,x4^2 - 40\sqrt{5}\,x2\,x3\,x4^2 - 2\,x3^2\,x4^2 - 5\sqrt{5}\,x3^2\,x4^2 -\\
&18\,x0\,x4^3 + 10\sqrt{5}\,x0\,x4^3 - 18\,x1\,x4^3 - 10\sqrt{5}\,x1\,x4^3 - 18\,x2\,x4^3 - 10\sqrt{5}\,x2\,x4^3 - 18\,x3\,x4^3 + 10\sqrt{5}\,x3\,x4^3 + 8\,x4^4\big)
\end{aligned}$$

「$x_0 + x_1 + x_2 + x_3 + x_4 = 0$」なので, 確かに式（2）は正しいです.

プログラム relation

上の長い操作を自動化したプログラムです.(評価してください)

In[1176]:=

```
Clear[ξ]
```

In[1177]:=

```
relation[bases_List,target_]:=Module[{vars,list,nbase,new,k},
vars={x4,x3,x2,x1,x0};
$target=Simplify[Collect[PolynomialMod[target,Cyclotomic[5,ξ]],ξ]/.{ξ^2→2Cos[4/5Pi]-ξ^3}];
polys=Append[bases,$target];
list=Simplify[PolynomialReduce[#,base,vars]〚2〛&/@polys];
k=1;
nbase={list〚1〛-m1};
Do[k++;new=Simplify[PolynomialReduce[list〚k〛,nbase,vars]〚2〛];
AppendTo[nbase,new-ToExpression["m"<>ToString[k]]],Length[bases]-1];
Expand@Simplify[PolynomialReduce[Last[list],nbase,vars]〚2〛]]
```

polyをbases(集合)で表したい時, 構文は *relation[bases(リスト), poly]* です. この例の場合は次の様になります.

In[1178]:=

```
relation[{d5, e5, d3 a2 , d2 a3} , poly2]
```

Out[1178]=

$$-\frac{125\ a2\ a3}{2}-\frac{375\ a5}{2}+\frac{125\ m1}{2}+125\ m2-\frac{25\ m3}{4}-\frac{75\ m4}{4}$$

base はグレブナー基底なので, bases の順序を変えても結果は変わりません.

In[1179]:=

```
relation[{ d3 a2 , e5, d5, d2 a3} , poly2]
```

Out[1179]=

$$-\frac{125\ a2\ a3}{2}-\frac{375\ a5}{2}-\frac{25\ m1}{4}+125\ m2+\frac{125\ m3}{2}-\frac{75\ m4}{4}$$

§4 − 3 $r_4{}^2 r_2 + r_1{}^2 r_3$ について

relation を使って証明します.

In[1180]:=

```
poly3 = r4^2 r2 + r1^2 r3;
relation[{d3}, poly3]
```

Out[1181]=

$$-\frac{25\,a3}{2}-\frac{5\,m1}{2}$$

これは poly3= $-\frac{25\,a3}{2}-\frac{5\,d_3}{2}$ を表します.

【検算】

In[1182]:=

```
poly3 = Simplify[
   Collect[PolynomialMod[poly3, Cyclotomic[5, ξ]], ξ] /. {ξ^2 → 2 Cos[4 / 5 Pi] - ξ^3}];
poly3 + 5 / 2 d3 + 25 / 2 $a3 // Factor
```

Out[1183]=

$$\frac{1}{2}\,(x0+x1+x2+x3+x4)\,\left(4\,x0^2-7\,x0\,x1+6\sqrt{5}\,x0\,x1+4\,x1^2-7\,x0\,x2-6\sqrt{5}\,x0\,x2-7\,x1\,x2+6\sqrt{5}\,x1\,x2+4\,x2^2-7\,x0\,x3-6\sqrt{5}\,x0\,x3-7\,x1\,x3-6\sqrt{5}\,x1\,x3-7\,x2\,x3+6\sqrt{5}\,x2\,x3+4\,x3^2-7\,x0\,x4+6\sqrt{5}\,x0\,x4-7\,x1\,x4-6\sqrt{5}\,x1\,x4-7\,x2\,x4-6\sqrt{5}\,x2\,x4-7\,x3\,x4+6\sqrt{5}\,x3\,x4+4\,x4^2\right)$$

「$x_0+x_1+x_2+x_3+x_4=0$」なので，確かに式（3）は正しいです.

§4 – $4\, r_1^3\, r_2 + r_4^3\, r_3$ について

In[1184]:=

```
poly4 = (r1^3 r2 + r4^3 r3 // PolynomialMod[#, Cyclotomic[5, ξ]] & // Collect[#, ξ] &) /.
    {ξ^2 → 2 Cos[4 / 5 Pi] - ξ^3} // FullSimplify
```

Out[1184]=

$$
\begin{aligned}
&\frac{1}{2} \Big(4\, x0^4 + 4\, x1^4 + 4\, x2^4 + 4\, x3^4 - 2 \left(2 + \sqrt{5}\right) x2^3\, x4 + 6 \left(-1 + \sqrt{5}\right) x2^2\, x4^2 - 2 \left(2 + \sqrt{5}\right) x2\, x4^3 + 4\, x4^4 + \\
&\quad 2 \left(-2 + \sqrt{5}\right) x3^3 (x2 + x4) + 2\, x0^3 \left(\left(-2 + \sqrt{5}\right) x1 - \left(2 + \sqrt{5}\right) x2 - \left(2 + \sqrt{5}\right) x3 + \left(-2 + \sqrt{5}\right) x4\right) + \\
&\quad x3 (x2 + x4) \left(2 \left(-2 + \sqrt{5}\right) x2^2 + \left(7 - 5 \sqrt{5}\right) x2\, x4 + 2 \left(-2 + \sqrt{5}\right) x4^2\right) - \\
&\quad 6\, x3^2 \left(\left(1 + \sqrt{5}\right) x2^2 - \left(3 + \sqrt{5}\right) x2\, x4 + \left(1 + \sqrt{5}\right) x4^2\right) + 2\, x1^3 \left(\left(-2 + \sqrt{5}\right) x2 - \left(2 + \sqrt{5}\right) (x3 + x4)\right) + \\
&\quad 3\, x1^2 \left(-2 \left(1 + \sqrt{5}\right) x2^2 + 2 \left(-1 + \sqrt{5}\right) x3^2 - 2 \left(-3 + \sqrt{5}\right) x3\, x4 + 2 \left(-1 + \sqrt{5}\right) x4^2 + x2 \left(x3 - \sqrt{5}\, x3 + x4 + \sqrt{5}\, x4\right)\right) + \\
&\quad 3\, x0^2 \Big(-2 \left(1 + \sqrt{5}\right) x1^2 + 2 \left(-1 + \sqrt{5}\right) x2^2 + 2 \left(-1 + \sqrt{5}\right) x3^2 - \left(-1 + \sqrt{5}\right) x3\, x4 - 2 \left(1 + \sqrt{5}\right) x4^2 + \\
&\qquad x2 \left(6\, x3 - 2 \sqrt{5}\, x3 + x4 + \sqrt{5}\, x4\right) + x1 \left(x2 - \sqrt{5}\, x2 + x3 + \sqrt{5}\, x3 + 6\, x4 + 2 \sqrt{5}\, x4\right)\Big) + \\
&\quad x1 \Big(2 \left(-2 + \sqrt{5}\right) x2^3 + 3\, x2^2 \left(2 \left(3 + \sqrt{5}\right) x3 + \left(1 + \sqrt{5}\right) x4\right) + 3\, x2 \left(-\left(\left(-1 + \sqrt{5}\right) x3^2\right) - 8\, x3\, x4 - 2 \left(-3 + \sqrt{5}\right) x4^2\right) - \\
&\qquad (x3 + x4) \left(2 \left(2 + \sqrt{5}\right) x3^2 - \left(7 + 5 \sqrt{5}\right) x3\, x4 + 2 \left(2 + \sqrt{5}\right) x4^2\right)\Big) + x0 \Big(2 \left(-2 + \sqrt{5}\right) x1^3 - 2 \left(2 + \sqrt{5}\right) x2^3 - \\
&\qquad 2 \left(2 + \sqrt{5}\right) x3^3 - 3 \left(-1 + \sqrt{5}\right) x3^2\, x4 + 6 \left(3 + \sqrt{5}\right) x3\, x4^2 + 2 \left(-2 + \sqrt{5}\right) x4^3 + 3\, x2^2 \left(x3 + \sqrt{5}\, x3 + 6\, x4 - 2 \sqrt{5}\, x4\right) + \\
&\qquad 3\, x1^2 \left(2 \left(3 + \sqrt{5}\right) x2 + x3 + \sqrt{5}\, x3 + x4 - \sqrt{5}\, x4\right) + 3\, x2 \left(\left(1 + \sqrt{5}\right) x3^2 - 8\, x3\, x4 + \left(1 + \sqrt{5}\right) x4^2\right) - \\
&\qquad 3\, x1 \left(\left(-1 + \sqrt{5}\right) x2^2 + 2 \left(-3 + \sqrt{5}\right) x3^2 + 8\, x3\, x4 + \left(-1 + \sqrt{5}\right) x4^2 + 8\, x2 (x3 + x4)\right)\Big)\Big)
\end{aligned}
$$

In[1185]:=

```
relation[{d4, e4, d2 a2}, poly4]
```

Out[1185]=

$$-\frac{5\, a2^2}{2} - 40\, a4 + \frac{25\, m1}{2} + \frac{15\, m2}{2} - \frac{15\, m3}{2}$$

これは「$poly4 = -\dfrac{5\, a2^2}{2} - 40\, a4 + \dfrac{25\, d_4}{2} + \dfrac{15\, e_4}{2} - \dfrac{15\, d_2\, a_2}{2}$」を表します.

【検算】

In[1186]:=

```
poly4 - (25 / 2 d4 + 15 / 2 e4 - 15 / 2 d2 (s2) - 40 s4 - 5 / 2 (s2)^2) // Factor
```

Out[1186]=

$$
\begin{aligned}
&(x0 + x1 + x2 + x3 + x4) \\
&\Big(2\, x0^3 - 4\, x0^2\, x1 + \sqrt{5}\, x0^2\, x1 - 4\, x0\, x1^2 + \sqrt{5}\, x0\, x1^2 + 2\, x1^3 - 4\, x0^2\, x2 - \sqrt{5}\, x0^2\, x2 - 8\, x0\, x1\, x2 + \\
&\quad 11 \sqrt{5}\, x0\, x1\, x2 - 4\, x1^2\, x2 + \sqrt{5}\, x1^2\, x2 - 4\, x0\, x2^2 - \sqrt{5}\, x0\, x2^2 - 4\, x1\, x2^2 + \sqrt{5}\, x1\, x2^2 + 2\, x2^3 - 4\, x0^2\, x3 - \\
&\quad \sqrt{5}\, x0^2\, x3 - 8\, x0\, x1\, x3 - 11 \sqrt{5}\, x0\, x1\, x3 - 4\, x1^2\, x3 - \sqrt{5}\, x1^2\, x3 - 8\, x0\, x2\, x3 - 11 \sqrt{5}\, x0\, x2\, x3 - \\
&\quad 8\, x1\, x2\, x3 + 11 \sqrt{5}\, x1\, x2\, x3 - 4\, x2^2\, x3 + \sqrt{5}\, x2^2\, x3 - 4\, x0\, x3^2 - \sqrt{5}\, x0\, x3^2 - 4\, x1\, x3^2 - \sqrt{5}\, x1\, x3^2 - \\
&\quad 4\, x2\, x3^2 + \sqrt{5}\, x2\, x3^2 + 2\, x3^3 - 4\, x0^2\, x4 + \sqrt{5}\, x0^2\, x4 - 8\, x0\, x1\, x4 + 11 \sqrt{5}\, x0\, x1\, x4 - 4\, x1^2\, x4 - \sqrt{5}\, x1^2\, x4 - \\
&\quad 8\, x0\, x2\, x4 - 11 \sqrt{5}\, x0\, x2\, x4 - 8\, x1\, x2\, x4 - 11 \sqrt{5}\, x1\, x2\, x4 - 4\, x2^2\, x4 - \sqrt{5}\, x2^2\, x4 - 8\, x0\, x3\, x4 + \\
&\quad 11 \sqrt{5}\, x0\, x3\, x4 - 8\, x1\, x3\, x4 - 11 \sqrt{5}\, x1\, x3\, x4 - 8\, x2\, x3\, x4 + 11 \sqrt{5}\, x2\, x3\, x4 - 4\, x3^2\, x4 + \sqrt{5}\, x3^2\, x4 - \\
&\quad 4\, x0\, x4^2 + \sqrt{5}\, x0\, x4^2 - 4\, x1\, x4^2 - \sqrt{5}\, x1\, x4^2 - 4\, x2\, x4^2 - \sqrt{5}\, x2\, x4^2 - 4\, x3\, x4^2 + \sqrt{5}\, x3\, x4^2 + 2\, x4^3\Big)
\end{aligned}
$$

「$x_0 + x_1 + x_2 + x_3 + x_4 = 0$」なので，確かに式（4）は正しいです.

§5. 解を求める

$$\begin{cases} r_1\, r_4 = -\frac{1}{2}\, d_2 - \frac{5}{2}\, a_2 & (1) \\ r_1{}^5 + r_4{}^5 = \frac{125}{2}\, d_5 + 125\, e_5 - \frac{25}{4}\, d_3\; a_2 - \frac{75}{4}\, d_2\; a_3 - \frac{125}{2}\, a_2\; a_3 - \frac{375}{2}\, a_5 & (2) \\ r_4{}^2\, r_2 + r_1{}^2\, r_3 = -\frac{5}{2}\, d_3 - \frac{25}{2}\, a_3 & (3) \\ r_1{}^3\, r_2 + r_4{}^3\, r_3 = \frac{25}{2}\, d_4 + \frac{15}{2}\, e_4 - \frac{15}{2}\, d_2\; a_2 - 40\, a_4 - \frac{5}{2}\, a_2{}^2 & (4) \end{cases}$$

§4より上の式が確かめられました．（1）（2）より $r_1{}^5$，$r_2{}^5$ は次の「R（x）= 0」の２解となります．

In[1187]:=

$$R := x\hat{}2 - \left(\frac{125}{2}\, d5 + 125\, e5 - \frac{25}{4}\, d3\; a2 - \frac{75}{4}\, d2\; a3 - \frac{125}{2}\, a2\; a3 - \frac{375}{2}\, a5\right) x + \left(-\frac{1}{2}\, d2 - \frac{5}{2}\, a2\right)^5$$

この2解を R_1，R_2 とすると，　$r_1 = \sqrt[5]{R_1}$，$r_4 = \sqrt[5]{R_2}$ です．
まず「$r_1 = r_4 = 0$」のとき $\left(F_1 = x^2 \text{ のとき}\right)$ は，τにより 解の名前を変える事により，
F_1 と F_2 を入れ替えます．このとき「$F_1 = F_2 = x^2$」とすると「$r_1 = r_2 = r_3 = r_4 = 0$」となり，
$x_k = 0$（$0 \le k \le 4$）となり矛盾します．
故に適当な解の名前の変換により「$r_1 \neq 0$」とできます． 以下「$r_1 \neq 0$」とします．
このときr_1 の5乗根はすべて異なりますが，どれをとってもfの解集合は同じになります．また
r_4 は「$r_1\, r_4 = -\frac{1}{2}\, d_2 - \frac{5}{2}\, a_2$」となるように累乗根を取ります．

次に（3）（4）は｛r_2，r_3｝についての一次方程式です．
それに対応する行列の行列式Detは「$\mathrm{Det} = r_4{}^5 - r_1{}^5$」です．
ここで「$\mathrm{Det} = 0$」とならないことが以下の様に証明されます．$\left(u_k \to r_k \text{ と変えています}\right)$

In fact, if $r1^5 = r4^5 \neq 0$, we have $r4 = \omega^i\ r1$ for some i, where ω is the primitive root of unit which has been chosen. Thus, if we denote by k1 and k3 the right hand sides of Equations 1 and 3 respectively, we have $r1^2 = k1 \big/ \omega^i$ and $r2\ \omega^2 + r3 = k3$. As $r2\ r3 = d2 / 2 - 5\ a5$ (conjugate equation of Equation (1)), we see that all the ui and thus all the roots belong to an extension of K of degree prime to 5, which implies that f is not irreducible.

しかしこの証明は私には納得できなかったので，ここでは説明できません．
（^^;）仮にこの証明が正しいとすると，（たぶんそうだと思いますが）r_2，
r_3，r_4 が全てr_1 により 1 通りに表せることになり，解の公式が完結します．

以下，§2で挙げたのと同じ「例3」の厳密解とその近似値を求めて，NSolveの数値解と比較します．

■ 【例】$f = 12 - 6x + 12x^2 - 9x^3 + x^5$ （$\mathrm{Gal} = F_{20}$）

In[1188]:=
```
f = 12 - 6 x + 12 x^2 - 9 x^3 + x^5;
```

(ア)f1, f2 と R を求めます.

In[1189]:=
```
{f1, f2} = findF12[f]
{fp, fm} = {f1 + f2, f1 - f2}
{a1, a2, a3, a4, a5} = CoefficientList[f, x, 5] // Reverse;
{e1, e2, e3, e4, e5} = CoefficientList[fp, x, 5] // Reverse;
{d1, d2, d3, d4, d5} = Sqrt[5] CoefficientList[fm, x, 5] // Reverse;
R
```

Out[1189]=

$$\left\{-93 - 15\sqrt{41} + \frac{171x}{2} + \frac{21\sqrt{41}\,x}{2} + 6x^2 - \frac{27x^3}{2} - \frac{3\sqrt{41}\,x^3}{2} + x^5,\right.$$
$$\left.-93 + 15\sqrt{41} + \frac{171x}{2} - \frac{21\sqrt{41}\,x}{2} + 6x^2 - \frac{27x^3}{2} + \frac{3\sqrt{41}\,x^3}{2} + x^5\right\}$$

Out[1190]=

$$\left\{-186 + 171x + 12x^2 - 27x^3 + 2x^5,\ -30\sqrt{41} + 21\sqrt{41}\,x - 3\sqrt{41}\,x^3\right\}$$

Out[1194]=

$$\left(\frac{45}{2} + \frac{3\sqrt{205}}{2}\right)^5 - \left(-18\,750 - 1200\sqrt{205}\right)x + x^2$$

（イ） R = 0 の解のうち0でない解をR_1とし，その5乗根r_1を求め，式（1）によりr_4も求めます.

In[1195]:=
```
{R1, R2} = x /. Solve[R == 0, x]
```

Out[1195]=

$$\left\{\frac{3}{4}\left(-12\,500 - 800\sqrt{205} - \sqrt{\frac{1}{2}\left(282\,490\,000 + 19\,577\,200\sqrt{205}\right)}\right),\right.$$
$$\left.\frac{3}{4}\left(-12\,500 - 800\sqrt{205} + \sqrt{\frac{1}{2}\left(282\,490\,000 + 19\,577\,200\sqrt{205}\right)}\right)\right\}$$

In[1196]:=
```
r1 = R1^(1 / 5) // FullSimplify
```

Out[1196]=

$$\left(-\frac{15}{2}\left(1250 + 80\sqrt{205} + \sqrt{1\,412\,450 + 97\,886\sqrt{205}}\right)\right)^{1/5}$$

In[1197]:=
```
r4 = (-1 / 2 d2 - 5 / 2 a2) / r1 // FullSimplify
```

Out[1197]=

$$-(-1)^{4/5}\left(9375 + 600\sqrt{205} - 15\sqrt{\frac{1}{2}\left(706\,225 + 48\,943\sqrt{205}\right)}\right)^{1/5}$$

（ウ） （3）（4）を使って r_2，r_3 を求めます.

In[1198]:=

```
{r2, r3} = 1 / (r4^5 - r1^5) {{r4^3, -r1^2}, {-r1^3, r4^2}}.
    {-5/2 d3 - 25/2 a3, 25/2 d4 + 15/2 e4 - 15/2 d2 a2 - 40 a4 - 5/2 a2^2} // FullSimplify
```

Out[1198]=

$$\left\{-(-1)^{2/5}\left(15\left(625-40\sqrt{205}+\sqrt{\frac{1}{2}\left(706225-48943\sqrt{205}\right)}\right)\right)^{1/5},\right.$$
$$\left.(-1)^{3/5}\left(9375-600\sqrt{205}-\frac{15}{2}\sqrt{1412450-97886\sqrt{205}}\right)^{1/5}\right\}$$

（エ）Lagrange　の逆変換を使い　解X = $\{x_0, x_1, x_2, x_3, x_4\}$ を求めます.
長いので　x_1 のみ表示します.

In[1199]:=

```
X = 1 / 5 {{1, 1, 1, 1, 1}, ξ^{0, 4, 3, 2, 1}, ξ^{0, 3, 1, 4, 2},
    ξ^{0, 2, 4, 1, 3}, ξ^{0, 1, 2, 3, 4}}.{0, r1, r2, r3, r4};
TraditionalForm[x1]
```

Out[1200]//TraditionalForm=

x1

(オ)「ζ= cos (2 π/5) + $\imath$ sin (2 π/5)」を代入して，近似値を求め，NSolveを使った数値解と比較します．御覧の様に一致します．

In[1201]:=

```
$X = {x0, x1, x2, x3, x4} = X /. {ξ → Cos[2 / 5 Pi] + I Sin[2 / 5 Pi]} // Simplify;
N[$X, 20] // Chop // Sort
```

Out[1202]=

```
{-3.6108679416295143647, -0.08922828766409501836 - 0.92456925884689160321 i,
 -0.08922828766409501836 + 0.92456925884689160321 i,
 1.8946622584788522007 + 0.5119200060193428023 i,
 1.8946622584788522007 - 0.5119200060193428023 i}
```

In[1203]:=

```
x /. NSolve[f == 0, x, 20]
```

Out[1203]=

```
{-3.6108679416295143647, -0.08922828766409501836 - 0.92456925884689160321 i,
 -0.08922828766409501836 + 0.92456925884689160321 i,
 1.8946622584788522007 - 0.5119200060193428023 i,
 1.8946622584788522007 + 0.5119200060193428023 i}
```

§6 原論文に載っているプログラム

With this definition, the MAPLE procedure is:

$\mathit{quintic} := \mathit{proc}\,(\mathit{pol});$

$x := \mathit{op}\,(\mathit{indets}\,(\mathit{pol}));$

$t := \mathit{coeff}(\mathit{pol}, x, 4)/\mathit{coeff}(\mathit{pol}, x, 5)/5;$

$f := \mathit{primpart}\,(\mathit{subs}\,(x{=}x{-}t, \mathit{pol}));$

$D := \mathit{discrim}\,(f, x);$

$\mathit{for}\quad i\quad \mathit{from}\quad 2\quad \mathit{to}\quad 5\quad \mathit{do}\quad a_i := \mathit{coeff}(f, x, 5-i)\quad \mathit{od};$

$f_{10} := \mathit{factor}\,(\ \langle\ \text{Equation 5}\ \rangle\ , D^{1/2});$

$\mathit{if\ not\ type}\,(f_{10}, `*`)\ \mathit{then\ RETURN}\,(\text{"the quintic is not solvable"})\ \mathit{fi};$

$(f_1, f_2) := \mathit{op}\,(f_{10});$

$e_4 := \mathit{coeff}(f_1 + f_2, s, 1);\quad e_5 := \mathit{coeff}(f_1 + f_2, s, 0);$

$d_2 := \sqrt{5}\cdot \mathit{coeff}(f_1 - f_2, s, 3);\quad d_3 := \sqrt{5}\cdot \mathit{coeff}(f_1 - f_2, s, 2);$

$d_4 := \sqrt{5}\cdot \mathit{coeff}(f_1 - f_2, s, 1);\quad d_5 := \sqrt{5}\cdot \mathit{coeff}(f_1 - f_2, s, 0);$

$\mathit{if}\quad d_2/2 - 5\,a_2/2 = 0$

$\qquad \mathit{and}\quad 125\,d_5/2 + 125\,e_5 - 25\,d_3\,a_2/4 - 75\,d_2\,a_3/4 - 125\,a_2\,a_3/2 - 375\,a_5/2 = 0$

$\qquad \mathit{then}\quad d_2 := -d_2;\quad d_3 := -d_3;\quad d_4 := -d_4;\quad d_5 := -d_5;\quad \mathit{fi};$

$h_1 := -d_2/2 - 5\,a_2;\quad h_3 := -5\,d_3/2 - 25\,a_3/2;$

$h_2 := 125\,d_5/2 + 125\,e_5 - 25\,d_3\,a_2/4 - 75\,d_2\,a_3/4 - 125\,a_2\,a_3/2 - 375\,a_5/2;$

$h_4 := 25\,d_4/2 + 15\,e_4/2 - 15\,d_2\,a_2/2 - 40\,a_4 - 5\,a_2^2/2;$

$\mathit{if}\quad h_1 = 0\quad \mathit{then}\quad u_1 := h_2^{1/5};\quad u_4 := 0;\quad d := h_2$

$\qquad \mathit{else}\quad d := \sqrt{h_2^2 - 4\,h_1^5};\quad u_1 := ((-h_2 + d)/2)^{1/5};\quad u_4 := h_1/u_1;\quad \mathit{fi};$

$\omega := (\sqrt{-2\sqrt{5} - 10} + \sqrt{5} - 1)/4;$

$u_2 := (h_4\,u_1^2 - h_3\,u_4^3)/d;\quad u_3 := (h_3\,u_1^3 - h_4\,u_4^2)/d;$

$\mathit{result} :=$

$\qquad -t + (u_1 + u_2 + u_3 + u_4)/5,$

$\qquad -t + (\omega\,u_1 + \omega^2\,u_2 + \omega^3\,u_3 + \omega^4\,u_4)/5,\quad -t + (\omega^2\,u_1 + \omega^4\,u_2 + \omega\,u_3 + \omega^3\,u_4)/5,$

$\qquad -t + (\omega^3\,u_1 + \omega\,u_2 + \omega^4\,u_3 + \omega^2\,u_4)/5,\quad -t + (\omega^4\,u_1 + \omega^3\,u_2 + \omega^2\,u_3 + \omega\,u_4)/5$

$\mathit{end};$

上のプログラムはメインプログラムで，この他にサブプログラムがいくつか必要です．なお，私は既にガロア分解式を使った5次方程式を解くプログラム (solveQuinticProgramVer2) をアップしてあり，それでもすでに1秒以内に解は求まるので，この短い解の公式 (FormulaA) を使ったプログラムは（少なくとも当分の間は）書かないと思います．このファイルではこの素晴らしい解の公式を紹介したかっただけです．なお公式の右辺に関して「なぜ d_i, e_i を使ったか？」という疑問が起こりますが，これに関しては「MolienSeries.nb」をご覧ください．また，この他にも解の公式(FormulaB)も作りました．そちらは FormulaB.nb をご覧ください．